

# HOWTO install multiple Sybase ASE or RS versions on a Windows server

---

## Contents

1	Introduction .....	2
2	Using srvany.exe .....	2
3	ASE SERVER .....	2
3.1	Files needed .....	2
3.2	To install.....	2
4	BACKUP SERVER .....	3
5	REPLICATION SERVER .....	3
5.1	Files needed .....	3
5.2	To install.....	3
6	Final Tidy Up .....	4
7	Appendix A - astart_15_7_sp122.cmd.....	5
8	Appendix B - repstart_15_2.cmd .....	6
9	Appendix C – shutdown.sql (ASE server).....	7
10	Appendix D – bckstart_15_7_sp122.cmd.....	7
11	Appendix E – shutdown_bck.sql .....	8
12	Appendix F – Instance specific start/stop scripts .....	8
12.1	ASE Start Script.....	9
12.2	BackupServer Start Script.....	9

## 1 Introduction

Why should different version of Sybase on a single Windows server be an issue? If you install multiple versions of Sybase ASE, the paths will become filled with contradictory references to both Sybase install locations. The Windows servers may or may not start but even if they do, you cannot be 100% sure which library paths you are using for the Sybase ASE server, so you could encounter support issues later down the line.

You can work around this by starting Sybase from a command window (MS-DOS), run the SYBASE.bat, followed by the RUN\_SERVER.bat. However, the ASE server will need this window to remain open for it to keep running and any accidental closure will cause the ASE server to shutdown.

The solution below shows how to run the ASE servers in the background, as a service.

## 2 Using srvany.exe

The srvany.exe is available as part of the NT resources kit and can be downloaded from the web. This Microsoft executable allows batch files to be installed as a service. More information on its use is available at <http://support.microsoft.com/kb/137890>.

## 3 ASE SERVER

### 3.1 Files needed

Example distribution files:

D:\startup\_scripts

```
asestart_15_7_sp122.cmd
asestart_SYBSRV1.cmd (calls asestart_15_7_sp122.cmd)
shutdown.sql
srvany.exe (can be downloaded from http://download.cnet.com/SrvAny/3000-2084\_4-10873628.html)
```

### 3.2 To install

1. Setup a batch file to run any Sybase ASE instances for the version you have installed. In this example, we are starting a ASE 15.7 sp122 server. An example of the file called asestart\_15\_7\_sp122.cmd is located in 7 Appendix A - asestart\_15\_7\_sp122.cmd.
2. Edit file asestart\_SYBSRV1.cmd and change the values at the top of the file to point to the correct server, username and password.

```
set SRVNAME=SYBSRV1
set ADMINUSER=sa
set ADMINPWD=somepassword
cmd /c "d:\startup_scripts\asestart_15_7_sp122 %SRVNAME% %1% %ADMINUSER% %ADMINPWD%"
```
3. Create directory D:\sybase\sybase\_15\_7\_sp122 and copy the SYBASE.bat file in to this directory.
4. Test the ASE server will now start from the asestart\_SYBSRV1.cmd, before proceeding.
5. Create the services

```
sc create "SYB_SYBSRV1" binpath= "d:\startup_scripts\srvany.exe
d:\startup_scripts\asestart_SYBSRV1.cmd"
DisplayName= "Sybase ASE Server - SYBSRV1"
```

6. Fix error if the service fails to start by manually hacking the registry!
  - a. add 'key'  
HKEY\_LOCAL\_MACHINE\SYSTEM\ControlSet001\Services\SYB\_SYBSRV1\Parameters
  - b. the add a 'String Value' under the new key  
Value name: Application  
Value data: D:\startup\_scripts\asestart\_SYBSRV1.cmd (don't forget to put the full path in)
  - c. edit the ImagePath entry to only read "d:\startup\_scripts\srvany.exe"
7. Start service

## 4 BACKUP SERVER

Use the same method for the ASE servers. Here are some examples commands:-

1. Create the service  
sc create "SYBBCK\_SYBSRV6\_BS" binpath= "d:\startup\_scripts\srvany.exe  
d:\startup\_scripts\bckstart\_SYBSRV6.cmd" DisplayName= "Sybase BCK Server - SYBSRV6\_BS"

Set the 'Log On' for the BS service as a domain account, if you want to backup databases to a network drive.

## 5 REPLICATION SERVER

### 5.1 Files needed

Example distribution file:

D:\startup\_scripts

repstart\_15\_2.cmd  
repstart\_ag\_rep\_15\_2.cmd  
repstart\_ag\_rep\_prod\_15\_2.cmd  
srvany.exe

### 5.2 To install

1. Setup a batch file to run any Sybase Replication instances for the version you have installed. In this example, we are starting a RS 15.2 server. An example of the file called repstart\_15\_2.cmd is located in 8 Appendix B - repstart\_15\_2.cmd.
2. Edit file repstart\_ag\_rep\_prod\_15\_2.cmd and change the values at the top of the file to point to the correct server, username and password.  
set SRVNAME=ag\_rep\_aegon\_prod\_RS  
set ADMINUSER=sa  
set ADMINPWD=somepassword  
cmd /c "D:\startup\_scripts\repstart\_15\_2 %SRVNAME% %1% %ADMINUSER% %ADMINPWD%"
3. Create the services  
sc create "REP\_ag\_rep\_aegon\_prod\_RS" binpath= "D:\startup\_scripts\srvany.exe  
D:\startup\_scripts\repstart\_ag\_rep\_prod\_15\_2.cmd" DisplayName= "Sybase RepServer - ag\_rep\_aegon\_PROD\_RS"
4. Fix error if the service fails to start by manually hacking the registry
  - a. add 'key'  
HKEY\_LOCAL\_MACHINE\SYSTEM\ControlSet001\Services\REP\_ag\_rep\_aegon\_prod\_RS\Parameters
  - b. the add a 'String Value' under the new key  
Value name: Application  
Value data: D:\startup\_scripts\repstart\_ag\_rep\_15\_2.cmd (don't forget to put the full path in)
  - c. edit the ImagePath entry to only read "d:\startup\_scripts\srvany.exe"

5. Start service

## 6 Final Tidy Up

Remove all automatically installed services, which were created as part of the installation of the Sybase binaries. These will only serve to confuse DBA not familiar with this setup.

Example commands:-

```
sc delete SYBSQL_SYBSRV1
sc delete SYBBCK_SYBSRV1_BS
```

## 7 Appendix A - asestart\_15\_7\_sp122.cmd

This is the file used to run ALL Sybase ASE servers for an installed version. In this example, the version is 15.0.3

Keep the version string length the same (e.g. 15\_7\_sp122) or modify the line "set version=%scriptname:~9,11%"

```
@echo off
REM *****
REM * Get contextual info
REM *****
set curdrive=%~d0
set curdir=%~dp0
set scriptname=%~n0%~x0
set version=%scriptname:~9,11%

REM *****
REM * Diagnostic info
REM *****
REM echo %curdrive%
REM echo %curdir%
REM echo %scriptname%
REM echo %version%
REM *****
cd /d "%curdir%"

REM *****
REM * Setup ENV for this version of Sybase
REM * and parse cmdline parameters.
REM *****
IF EXIST %curdrive%\SYBASE\sybase_%version%\SYBASE.bat. (
    call %curdrive%\SYBASE\sybase_%version%\SYBASE.bat
) ELSE (
    echo SYBASE.bat script not found.
    echo.
    echo %scriptname% expects to be run from
%curdrive%\SYBASE\startup_scripts
    echo and to find a SYBASE.bat file in
%curdrive%\SYBASE\sybase_%version%\
    goto :error
)

if "%1"=="". (
    echo SYBASE instance name not specified.
    goto :error
)

if "%2"=="". (
    echo No verb specified for service control.
    goto :error
)

if "%2"=="start" goto :start
if "%2"=="stop" goto :stop
REM *****
REM * SIMULATE RESTART WITH STOP FOLLOWED BY RESTART.
REM *****

if "%2"=="restart" goto :stop
```

```

REM *****
REM * DEFAULT TO START IN THE ABSENCE OF ANY VERB.
REM *****
goto :start

REM *****
REM * STOP VERB PROCESSING
REM *****
:stop
echo stopping %1.....
net stop ASE_%1
isql -S%1 -U%3 -P%4 -i%curdir%\shutdown.sql
if "%2"=="restart" goto :start
goto :EOF

REM *****
REM * START VERB PROCESSING
REM *****
:start
echo starting %1.....
start /B %SYBASE%\%SYBASE_ASE%\install\RUN_%1.bat
goto :EOF

REM *****
REM * Usage/Error PROCESSING
REM *****
:error
echo.
echo "Usage: %scriptname% instancename [start|stop|restart]"

:EOF

```

## 8 Appendix B - repstart\_15\_2.cmd

```

call D:\rep_15_2_esd3\SYBASE.bat
@echo off

if "%2"=="start" goto :start
if "%2"=="stop" goto :stop
if "%2"==" " goto :error

:start
echo starting Replication Server %1.....
start /B %SYBASE%\REP-15_2\install\%1.bat
goto :EOF

:stop
echo stopping Replication Server %1.....
net stop %1
isql -S%1 -U%3 -P%4 -iD:\startup_scripts\shutdown.sql
goto :EOF

:error
echo "Usage: <startup script> [start|stop]"

```

```
:EOF
```

## 9 Appendix C – shutdown.sql (ASE server)

```
shutdown with wait = "00:01:40"  
go
```

## 10 Appendix D – bckstart\_15\_7\_sp122.cmd

```
@echo off  
REM *****  
REM * Get contextual info  
REM *****  
  
set curdrive=%~d0  
set curdir=%~dp0  
set scriptname=%~n0%~x0  
set version=%scriptname:~9,11%  
  
REM *****  
REM * Diagnostic info  
REM *****  
REM echo %curdrive%  
REM echo %curdir%  
REM echo %scriptname%  
REM echo %version%  
REM *****  
  
cd /d "%curdir%"  
  
REM *****  
REM * Setup ENV for this version of Sybase  
REM * and parse cmdline parameters.  
REM *****  
IF EXIST %curdrive%\SYBASE\sybase_%version%\SYBASE.bat. (  
    call %curdrive%\SYBASE\sybase_%version%\SYBASE.bat  
) ELSE (  
    echo SYBASE.bat script not found.  
    echo.  
    echo %scriptname% expects to be run from  
%curdrive%\SYBASE\startup_scripts  
    echo and to find a SYBASE.bat file in  
%curdrive%\SYBASE\sybase_%version\  
    goto :error  
)  
  
if "%1"=="". (  
    echo SYBASE instance name not specified.  
    goto :error  
)  
  
if "%2"=="". (  
    echo No verb specified for service control.  
    goto :error  
)  
  
if "%2"=="start" goto :start
```

```

if "%2"=="stop" goto :stop
REM *****
REM * SIMULATE RESTART WITH STOP FOLLOWED BY RESTART.
REM *****

if "%2"=="restart" goto :stop

REM *****
REM * DEFAULT TO START IN THE ABSENCE OF ANY CONFIG.
REM *****
goto :start

REM *****
REM * STOP VERB PROCESSING
REM *****
:stop
echo stopping %1.....
net stop ASE_%1
isql -S%1 -U%3 -P%4 -i%curdir%\shutdown.sql
if "%2"=="restart" goto :start
goto :EOF

REM *****
REM * START VERB PROCESSING
REM *****
:start
echo starting %1.....
start /B %SYBASE%\%SYBASE_ASE%\install\RUN_%1_BS.bat
goto :EOF

REM *****
REM * Usage/Error PROCESSING
REM *****
:error
echo.
echo "Usage: %scriptname% instancename [start|stop|restart]"

:EOF

```

## 11 Appendix E – shutdown\_bck.sql

```

shutdown SYB_BACKUP
go

```

## 12 Appendix F – Instance specific start/stop scripts

The following are the scripts that would be invoked to start|stop|restart the installed instance service(s).



## 12.1 ASE Start Script

This instance would be named “dt6\_sybstart\_15\_7\_sp122.cmd”:

```
rem
rem      sybstart_dr1_15_7_sp122.bat
rem
rem
```

```
set SRVNAME=dt6_15_7_sp122
set ADMINUSER=sa
set ADMINPWD=(sa password)
```

```
cmd /c "D:\SYBASE\startup_scripts\sybstart_15_7_sp122 %SRVNAME% %1%
%ADMINUSER% %ADMINPWD%"
```

## 12.2 BackupServer Start Script

This instance would be named “dt6\_bckstart\_15\_7\_sp122.cmd”:

```
@echo off
set SRVNAME=dr1_15_7_sp122
set ADMINUSER=sa
set ADMINPWD=(sa password)
```

```
cmd /c "D:\SYBASE\startup_scripts\bckstart_15_7_sp122 %SRVNAME% %1%
%ADMINUSER% %ADMINPWD%"
```